```
(* Clear variables and define initial values.  h is the height of the center of the
 wheel, radius is the radius of the wheel, g is the acceleration due to gravity,
θ is the angle measured with respect to the "noon" line,
delta is the step size in angle,
tsol is the time it takes for the person to hit the water after stepping off the wheel,
xrange is the horizontal distance from the center of the wheel to the
 person's landing point in the water, x0 is the horizontal position of the
 person with respect to the center of the wheel when stepping off the wheel,
vboat is the constant speed of the boat, xboat is the horizontal position of the boat,
with respect to the center of the wheel, and variation is the distance
 between the boat and the person at the moment the person hits the water *)

Clear[radius, t, h, w, g, theta, delta, tsol, xrange, x0, dist, vboat, xboat, variation]
radius = 30;
h = 80;
w = 0.2; g = 9.81; theta[0] = 0;
delta = 0.001;
dist = 150; vboat = 10;
theta[n_] := theta[n] = theta[n - 1] + delta

(* x0[n] calculates the x value of the person on the wheel for each value of θ *)

x0[n_] := x0[n] = radius Sin[theta[n]]

(* The next step is
 critical:  We calculate the time it takes for the person to reach the water
    once stepping off the wheel.  We do this by solving the vertical equation
    of motion.  The initial height above the water is h + radius*cos θ,
the initial velocity is - w* radius * sin θ,
and -1/2 g t² is the familiar term taking the acceleration
  of gravity into account.  This quadratic will yield two solutions,
a negative and a positive.  Since we want the positive solution,
and knowing that Mathematica orders solutions in increasing numerical order,
we want to choose the second solution for future calculations. The
 notation [[2]] is the instruction to select the second solution
 to the equation. The statement "tsol[n]=t/. Solve... "  takes the
 positive solution of the quadratic, sets t equal to this value,
and we define tsol[n] to equal this value of t for the nth iteration. *)

tsol[n_] :=
 tsol[n] = t /. Solve[h + radius Cos[theta[n]] - w radius Sin[theta[n]] t - g t^2 / 2 == 0, t][[2]]

(* The next two statements compute the x position of the person and of the
 boat at the time of the persons's impact in the water.  For the person,
we write the equation of horizontal motion as
 "final horizontal position = initial horizontal position on the wheel + horizontal
   speed x time of flight.  For the boat, the time of travel we must use is the
   total time, i.e., the time the person was on the wheel plus the time the person
   was in flight.  The time of flight is tsol[n], the time on the wheel is simply
   theta[n]/w  Variation is the distance between the position of the boat and the
   position of the person upon impact; if xboat> xrange, then variation >0.  *)
```

```
xrange[n_] := xrange[n] = x0[n] + w radius Cos[theta[n]] tsol[n]
xboat[n_] := xboat[n] = dist - vboat (tsol[n] + theta[n] / w)
variation[n_] := variation[n] = xboat[n] - xrange[n]

(* The next statement sets up a condition for finding the first
 iteration when the person's landing position is greater than the
 position of the front of th eboat. One way to think about this condition
 in a recursive calcualtion is to realize that as the boat approaches,
variation[n]>0.  The first iteration in which the person lands on the boat, variation[n]<
 0.  This If statement determines the value of n for which that ratio changes sign,
indicating the person has landed in the boat.  The output consists
 of the value of the iteration, the angle, the range of the person,
the range of the boat, and the total time elapsed since t=theta =0.   *)

Catch[Do[If[Sign[variation[n] / variation[n - 1]] < 0,
    Throw[{n, theta[n], xrange[n], xboat[n], (theta[n] / w) + tsol[n]}]], {n, 1, 4000}]] // N

(*The next two statements use more obvious If statements to determine the
 window of success.  The If statement tests to determine the value of n such
 that the landing position of the person is for the frist time greater than
 the position of the front of the boat.  We declare that this corresponds
 to the beginning of the window of success.  The next If statement tests
 to determine the value of the iteration when the landing position of the
 person first exceeds the position of the front of the boat by 1 meter,
and we thus conclude that the last successful values are the m-1 iteration.  Notice
  that the two methods for determining first success yield the same answer.  *)


firstsol = Catch[Do[If[(xrange[p] > xboat[p]),
     Throw[{p, theta[p], xboat[p], xrange[p], theta[p] / w + tsol[p]}]], {p, 4000}]];
Print["The beginning of successful launch window occurs when theta =  ",
 180 firstsol[[2]] / π, " degrees."]
Print["The person steps off the wheel at t = ", firstsol[[2]] / w,
 "   seconds and hits the water at t =  ", firstsol[[5]], "  seconds."]
Print["  When the person hits the water, the front edge of the boat is at x = ",
 firstsol[[3]], "  meters."]
lastsol = Catch[Do[If[(xrange[m] - xboat[m]) > 1, Throw[{m - 1, theta[m - 1], xboat[m - 1],
        xrange[m - 1], theta[m - 1] / w + tsol[m - 1], w radius Cos[theta[m - 1]]}]], {m, 4000}]];
Print["   "]
Print["The successful launch window ends when theta = ",
 (180 / π) lastsol[[2]], " degrees."]
Print["The person steps off the wheel at ", lastsol[[2]] / w,
 " seconds and hits the water at ", lastsol[[5]], " seconds."]
Print["When the person hits the water, the front edge of the boat is at x = ",
 lastsol[[3]], " meters."]
g1 = ListPlot[Table[{tsol[n] + theta[n] / w, xboat[n]}, {n, 0, 4000}], PlotStyle → Red];
g2 = ListPlot[Table[{tsol[n] + theta[n] / w, xrange[n]}, {n, 4000}], PlotStyle → Cyan];

Show[g1, g2, PlotRange → All, AxesLabel → {Time, Distance },
 LabelStyle → Directive[Blue, 14, Bold]]
```

Out[844]= {1924., 1.924, 21.4297, 21.4288, 12.8571}

The beginning of successful launch window occurs when theta = 110.237 degrees.

The person steps off the wheel at t = 9.62
seconds and hits the water at t = 12.8571 seconds.

When the person hits the water, the front edge of the boat is at x = 21.4288 meters.

The successful launch window ends when theta = 113.675 degrees.

The person steps off the wheel at 9.92 seconds and hits the water at 13.1239 seconds.

When the person hits the water, the front edge of the boat is at x = 18.7614 meters.

Out[856]=